

Appendix B

Converting from PC/370 to MVS/ESA

Objectives

Upon completion of this chapter you will be able to:

- Describe the differences between the DOS commands used with PC/370 and the JCL commands used with MVS/ESA,
- Describe the use of ANSI carriage control characters, and
- Convert a PC/370 program to MVS/ESA.

Introduction

Two programs from this text, `TEACH2A.MLC` and `TEACH8A.MLC`, will be used to demonstrate the steps necessary to convert source code from PC/370 to assembly language for MVS/ESA compatible machines.

* * * * *

First, we will look at `TEACH2A.MLC`. Recall that this program lists the teacher records. There are no page headings in this program. The changes needed to assemble this program on MVS/ESA compatible machines are trivial. The new listing is on the next page. All changes have been highlighted. Several of the changes apply to internal documentation and `WTOS` only, and so will not be discussed. Most changes relate to file processing and include the `OPEN` macros, `DCB` macros, and record layouts.

The `OPEN` Macro

Keep in mind that when working within the MVS/ESA environment, all data is (typically) stored in EBCDIC format. Consequently, the `OI` (Or Immediate) instruction, with which we have always preceded the `OPEN`, is no longer used. The two `OI` instructions used in the PC/370 version of this program have been dropped. The format of the `OPEN` macro itself is different: you must indicate the `DCB` name *as well as the file mode*. For example:

```
OPEN (TEACHERS , INPUT)
OPEN (REPORT , OUTPUT)
```

The `DCB` Macro

The `DCB` macro for MVS/ESA is actually simpler than in PC/370. First, we add `DSORG=PS` (data set organization is physical sequential). Optionally, we remove the `RECFM` and `LRECL` parameters as these are typically specified in the JCL (Job Control Language, shown later) or taken from the system catalog entries for the target data sets at run time. (These parameters can be kept in the program provided they do not conflict with the values obtained from the JCL or system catalog. If they conflict, then a run time error is likely.) The `MACRF` parameter is changed from `MACRF=G` to

**APPENDIX B
CONVERTING FROM PC/370 TO MVS/ESA**

B.2

MACRF=GM (get move, for input) or from MACRF=P to MACRF=PM (put move, for output). The DDNAME parameter, rather than giving a DOS file specification, indicates the associated DDNAME on the JCL. The DDNAME is not in apostrophes. For example:

```

PRINT NOGEN
*****
*      FILENAME: B1973BQ.TRAINING.ASM (TEACH2A)      *
*      AUTHOR   : Bill Qualls                          *
*      SYSTEM   : IBM MVS/ESA Compatible             *
*      REMARKS  : A quick-and-dirty list of teachers.   *
*****
      START 0
      REGS
BEGIN   BEGIN
      WTO   'TEACH2A ... Begin execution'
      OPEN  (TEACHERS, INPUT)
      OPEN  (REPORT, OUTPUT)
LOOP    GET   TEACHERS,IREC      Read a single teacher record
      MVC   OTID,ITID           Move teacher ID Nbr to output
      MVC   OTNAME,ITNAME       Move teacher Name to output
      MVC   OTDEG,ITDEG        Move highest degree to output
      MVC   OTTEN,ITTEN        Move tenure to output
      MVC   OTPHONE,ITPHONE     Move phone nbr to output
      PUT   REPORT,OREC        Write report line
      B     LOOP
*
*      EOJ processing
*
ATEND   CLOSE TEACHERS
      CLOSE REPORT
      WTO   'TEACH2A ... Teacher list on //REPORT'
      WTO   'TEACH2A ... Normal end of program'
      RETURN
*
*      Literals, if any, will go here
*
      LTORG
*
*      File definitions
*
TEACHERS DCB DSORG=PS,MACRF=GM,EODAD=ATEND,DDNAME=TEACHERS
REPORT   DCB DSORG=PS,MACRF=PM,DDNAME=REPORT
*
*      Miscellaneous field definitions
*
*
*      Input record definition
*
IREC     DS      0CL80           Teacher record
ITID     DS      CL3             Teacher ID nbr
ITNAME   DS      CL15           Teacher name
ITDEG    DS      CL4             Highest degree
ITTEN    DS      CL1             Tenured?
ITPHONE  DS      CL4             Phone nbr
DS      CL53

```

(continued)

```

*
*      Output (line) definition
*
OREC  DS  0CL60
OTID    DS    CL3           Teacher ID nbr
        DC    CL3' '
OTNAME  DS    CL15          Teacher name
        DC    CL3' '
OTDEG   DS    CL4           Highest degree
        DC    CL3' '
OTTEN   DS    CL1           Tenured?
        DC    CL3' '
OTPHONE DS    CL4           Phone nbr
        DC  CL21' '
        END    BEGIN

```

No More CR/LF

The carriage return/line feed which we have had to accommodate in most of our PC/370 programs is *not* used in MVS/ESA. It is a PC consideration only, and so it is dropped from the program. All of our programs have used fixed length records. Input files of the type we have used are typically stored as members of a PDS (Partitioned Data Set), the mainframe equivalent to a PC directory. All members of the same PDS have the same record length, often 80. The input record layout has been changed to reflect its new length:

IREC	DS	0CL80	Teacher record
ITID	DS	CL3	Teacher ID nbr
ITNAME	DS	CL15	Teacher name
ITDEG	DS	CL4	Highest degree
ITTEN	DS	CL1	Tenured?
ITPHONE	DS	CL4	Phone nbr
	DS	CL53	

The output will usually *not* be written to a PDS, but rather to a sequential file on disk or tape, or to a print spool where it will be held (and may be viewed) until it is released for printing or is purged. The output record in this program has been changed to remove the CR/LF and have a total record length at 60:

OREC	DS	0CL60	
OTID	DS	CL3	Teacher ID nbr
	DC	CL3' '	
OTNAME	DS	CL15	Teacher name
	DC	CL3' '	
OTDEG	DS	CL4	Highest degree
	DC	CL3' '	
OTTEN	DS	CL1	Tenured?
	DC	CL3' '	
OTPHONE	DS	CL4	Phone nbr
	DC	CL21' '	

Be aware that the REGS, BEGIN, and RETURN macros are installation specific. Your company will have equivalent macros already defined, perhaps even with the same name.

Job Control Language

In the MVS/ESA world, JCL (Job Control Language) is used to run batch jobs. (Recall that batch jobs are those jobs which run unattended, as opposed to online programs which respond to a user's commands entered at a terminal.) The assembly and link steps are run in a batch mode, just as they are in PC/370. The following JCL will assemble and link program `TEACH2A`:

```
//B1973BQX JOB (1973,B1973BQ), 'BILL QUALLS',
//          CLASS=A, NOTIFY=B1973BQ, MSGCLASS=X
//*
//ASM      EXEC PGM=ASMA90, PARM='NODECK, OBJECT'
//SYSLIB   DD DSN=B1973BQ.TRAINING.MACLIB, DISP=SHR
//          DD DSN=SYS1.MACLIB, DISP=SHR
//SYSUT1   DD DSN=SYSUT1, SPACE=(1024, (120, 120)), UNIT=VIO
//SYSLIN   DD DSN=&&OBJ, DISP=(, PASS), SPACE=(CYL, (5, 1)), UNIT=VIO,
//          DCB=(BLKSIZE=3040, LRECL=80, RECFM=FBS, BUFNO=1)
//SYSPRINT DD SYSOUT=*
//SYSIN    DD DSN=B1973BQ.TRAINING.ASM(TEACH2A), DISP=SHR
//*
//LINK     EXEC PGM=HEWL, COND=(4, LT, ASM)
//SYSLIN   DD DSN=&&OBJ, DISP=(OLD, DELETE)
//SYSLMOD  DD DSN=B1973BQ.TRAINING.LOADLIB(TEACH2A), DISP=SHR
//SYSLIB   DD DSN=SYS1.LINKLIB, DISP=SHR
//SYSUT1   DD DSN=SYSUT1, SPACE=(1024, (120, 120), , , ROUND), UNIT=VIO
//SYSPRINT DD SYSOUT=*
```

The format of the `JOB` card is installation specific: yours will probably differ slightly. The program name for the assemble and link steps (`ASMA90` and `HEWL` above) will also likely differ. The file (or PDS and member) containing the source code is specified on the `SYSIN DD` of the assembly step. The file (or PDS and member) to contain the load module is specified on the `SYSLMOD DD` of the link step. The JCL can be likened to the steps we have used to link and execute our PC/370 programs as follows:

	PC/370	MVS/ESA
Step(s)	m370, a370	//ASM EXEC PGM=ASMA90
Source file (input)	TEACH2A.MLC	B1973BQ.TRAINGING.ASM(TEACH2A)
Assembly listing (output)	TEACH2A.PRN	SYSOUT=*
Object code (output)	TEACH2A.OBJ	&&OBJ (a temporary data set)
Step	l370	//LINK EXEC PGM=HEWL
Object code (input)	TEACH2A.OBJ	&&OBJ
Executable code (output)	TEACH2A.COM	B1973BQ.TRAINING.LOADLIB(TEACH2A)

In order to execute the program, JCL is used to assign each `DDNAME` (from the `DCB` macro in the program) to a file or device. This is not unique to assembly language programming: a `DD` entry would be used for each `ASSIGN TO` clause in a COBOL program. The following JCL will execute program `TEACH2A`:

APPENDIX B CONVERTING FROM PC/370 TO MVS/ESA

B.5

```
//B1973BQA JOB (1073,B1973BQ), 'BILL QUALLS',  
//          CLASS=A, NOTIFY=B1973BQ, MSGCLASS=X  
//TEACH2A EXEC PGM=TEACH2A  
//STEPLIB DD DSN=B1973BQ.TRAINING.LOADLIB, DISP=SHR  
//TEACHERS DD DSN=B1973BQ.TRAINING.DATA (TEACHER), DISP=SHR  
//REPORT DD SYSOUT=*, DCB=(RECFM=FB, LRECL=60, BLKSIZE=0)
```

The `STEPLIB` uses as its `DSN` the same `DSN` as the `//SYSLMOD` of the `LINK` step. The member name is not included, as it is taken from the `PGM` name.

The `DCB` parameters are typically specified for output files only: for input files these values are taken from the system catalog or file header. When the `DCB` parameters are given, for fixed length records (`RECFM=FB`) the block size (`BLKSIZE`) must be a multiple of the record length (`LRECL`). In some installations, the operating system will determine the optimal `BLKSIZE` for the specified device if `BLKSIZE=0` is used.

To the new user, `JCL` may seem cumbersome. But it does have some advantages over the methods used in `PC/370`. In particular, with `PC/370`, if we want to run the same program with a different file, we must change the filename on the `DDNAME` parameter of the `DCB`, reassemble, relink, and execute. With `JCL`, we change the filename (`DSN`) on the `JCL` only: the program is unchanged.

ANSI Carriage Control

The next program we will look at is `TEACH8A`. Like `TEACH2A`, this program produces a list of the teachers. However, this program also includes headings and page break logic. When a program produces a report with headings and page break logic, there are additional changes that need to be made to the program, above and beyond those shown in `TEACH2A`.

These changes have to do with how mainframe printers control form feed and line spacing. Mainframe printers typically make use of ANSI carriage control. According to this standard, the first position of the output record is reserved for the carriage control character as follows:

First position of record	Advance (before printing)
␣ (blank)	One line (single spacing)
0 (zero)	Two lines (double spaced)
- (hyphen)	Three lines (triple spaced)
+ (plus)	Zero lines (overwrite; suppress spacing)
1 (one)	To top of page

The `DCB` parameter `RECFM=FBA` indicates ANSI carriage control is being used.

The new listing is on the next page. All changes have been highlighted. Notice that all references to `CR/LF` have been removed. The first position of each output record has been reserved for the carriage control character. All references to `FORMFEED` have been removed, and the carriage

**APPENDIX B
CONVERTING FROM PC/370 TO MVS/ESA**

B.6

control character for HD1 is '1' indicating advance to top of page before printing. HD2 was originally used to provide a blank line between HD1 and HD3. It could have been retained, but instead it has been removed and the carriage control character for HD3 is '0' indicating advance two lines before printing.

Note: Mainframe reports are typically 133 characters wide, with the first byte reserved for the carriage control character and the remaining 132 bytes for the data.

```

PRINT NOGEN
*****
*      FILENAME:  B1973BQ.TRAINING.ASM(TEACH8A)      *
*      AUTHOR   :  Bill Qualls                       *
*      SYSTEM   :  IBM MVS/ESA Compatible            *
*      REMARKS  :  This is a revision of TEACH2C.MLC. *
*                  Produce list of teachers, with headings. *
*                  Introduce page break logic.         *
*****
      START 0
      REGS
BEGIN  BEGIN
      WTO   'TEACH8A ... Begin execution'
      BAL   R10,SETUP
MAIN   EQU   *
      CLI   EOFSW,C'Y'
      BE    EOJ
      BAL   R10,PROCESS
      B     MAIN
EOJ    EQU   *
      BAL   R10,WRAPUP
      WTO   'TEACH8A ... Normal end of program'
      RETURN
*****
*      SETUP - Those things which happen one time only, *
*              before any records are processed.        *
*****
SETUP  EQU   *
      ST   R10,SVSETUP
      OPEN (TEACHERS,INPUT)
      OPEN (REPORT,OUTPUT)
      BAL   R10,READ
      L     R10,SVSETUP
      BR    R10
*****
*      HDGS - Print headings.                            *
*****
HDGS   EQU   *
      ST   R10,SVHDGS
      AP   PGS,=P'1'           Add 1 to page count
      MVC  HDPGS,=X'40202120' Edit pattern for page count
      ED   HDPGS,PGS          Move page count to heading
      PUT  REPORT,HD1         PUT REPORT,FORMFEED dropped
      PUT  REPORT,HD3         PUT REPORT,HD2 dropped

```

(continued)

**APPENDIX B
CONVERTING FROM PC/370 TO MVS/ESA**

B.7

```

                PUT    REPORT,HD4
                ZAP    LNS,=P'0'           Reset line count to zero
                L      R10,SVHDGS
                BR     R10
*****
*              PROCESS - Those things which happen once per record.  *
*****
PROCESS EQU    *
                ST     R10,SVPROC
                BAL    R10,FORMAT
                BAL    R10,CHKLNS
                BAL    R10,WRITE
                BAL    R10,READ
                L      R10,SVPROC
                BR     R10
*****
*              READ - Read a record.  *
*****
READ EQU      *
                ST     R10,SVREAD
                GET    TEACHERS,IREC      Read a single teacher record
                B      READX
ATEND EQU     *
                MVI    EOFSW,C'Y'
READX EQU    *
                L      R10,SVREAD
                BR     R10
*****
*              FORMAT - Format a single detail line.  *
*****
FORMAT EQU    *
                ST     R10,SVFORM
                MVC    OTID,ITID          Move teacher ID Nbr to output
                MVC    OTNAME,ITNAME      Move teacher Name to output
                MVC    OTDEG,ITDEG        Move highest degree to output
                MVC    OTTEN,ITTEN        Move tenure to output
                MVC    OTPHONE,ITPHONE    Move phone nbr to output
                L      R10,SVFORM
                BR     R10
*****
*              CHKLNS - Check lines printed. Full page?  *
*****
CHKLNS EQU    *
                ST     R10,SVCHKLNS
                CP     LNS,MAXLNS
                BL     CHKLNSX
                BAL    R10,HDGS
CHKLNSX EQU   *
                L      R10,SVCHKLNS
                BR     R10
*****
*              WRITE - Write a single detail line.  *
*****
WRITE EQU    *
                ST     R10,SVWRITE
                PUT    REPORT,OREC        Write report line
                AP     LNS,=P'1'
                L      R10,SVWRITE
                BR     R10

```

(continued)

**APPENDIX B
CONVERTING FROM PC/370 TO MVS/ESA**

```

*****
*          WRAPUP - Those things which happen one time only,      *
*                   after all records have been processed.        *
*****
WRAPUP  EQU      *
        ST      R10,SVWRAP
        CLOSE  TEACHERS
        CLOSE  REPORT
WTO    'TEACH8A ... Teacher list on //REPORT'
        L      R10,SVWRAP
        BR     R10
*****
*          Literals, if any, will go here                          *
*****
        LTORG
*****
*          File definitions                                        *
*****
TEACHERS DCB  DSORG=PS,MACRF=GM,EODAD=ATEND,DDNAME=TEACHERS
REPORT  DCB  DSORG=PS,MACRF=PM,DDNAME=REPORT
*****
*          RETURN ADDRESSES                                        *
*****
SVSETUP DC      F'0'          SETUP
SVHDGS  DC      F'0'          HDGS
SVPROC  DC      F'0'          PROCESS
SVREAD  DC      F'0'          READ
SVFORM  DC      F'0'          FORMAT
SVWRITE DC      F'0'          WRITE
SVWRAP  DC      F'0'          WRAPUP
SVCHKLNS DC     F'0'          CHKLNS
*****
*          Miscellaneous field definitions                          *
*****
EOFSW   DC      CL1'N'        End of file? (Y/N)
PGS     DC      PL2'0'        Nbr of pages printed.
LNS     DC      PL2'3'        Lines printed on this page.
MAXLNS  DC      PL2'3'        Max nbr lines per page.
*          My line counts exclude hdgs.
*****
*          Input record definition                                *
*****
IREC    DS      0CL80          Teacher record
ITID    DS      CL3           Teacher ID nbr
ITNAME  DS      CL15          Teacher name
ITDEG   DS      CL4           Highest degree
ITTEN   DS      CL1           Tenured?
ITPHONE DS      CL4           Phone nbr
DS      CL53
*****
*          Output (line) definition                              *
*****
OREC    DS      0CL60
OCC     DC      CL1' '        Carriage Control
OTID    DS      CL3           Teacher ID nbr
        DC      CL3' '
OTNAME  DS      CL15          Teacher name
        DC      CL3' '
OTDEG   DS      CL4           Highest degree
        DC      CL3' '

```

(continued)

**APPENDIX B
CONVERTING FROM PC/370 TO MVS/ESA**

B.9

```

OTTEN   DS      CL1              Tenured?
        DC      CL3'  '
OTPHONE DS      CL4              Phone nbr
        DC      CL20'  '
*
*      Headings definitions
*
HD1   DS    0CL60
        DC    CL1'1'
        DC      CL40'              LIST OF TEACHERS      Page'
HDFGS   DC      CL4'BZZ9'
        DC      CL15'  '
HD3   DS    0CL60
        DC    CL1'0'
        DC      CL40'ID#          Name          Degr  Ten  Phone'
        DC      CL19'  '
HD4   DS    0CL60
        DC    CL1'  '
        DC      CL40'----  -----  ---  ---  ----'
        DC      CL19'  '
        END      BEGIN

```

The JCL to execute program TEACH8A is as follows. Note the RECFM=FBA parameter indicating that ANSI carriage control is being used.

```

//B1973BQX JOB (1973,B1973BQ), 'BILL QUALLS',
//          CLASS=A, NOTIFY=B1973BQ, MSGCLASS=X
//TEACH8A EXEC PGM=TEACH8A
//STEPLIB DD DSN=B1973BQ.TRAINING.LOADLIB, DISP=SHR
//TEACHERS DD DSN=B1973BQ.TRAINING.DATA(TEACHER), DISP=SHR
//REPORT DD SYSOUT=*, DCB=(RECFM=FBA, LRECL=60, BLKSIZE=0)

```

The result will appear as follows when printed. The carriage control characters are recognized by the printer but are not shown on the report.

```

                LIST OF TEACHERS                Page 1
ID#           Name           Degr  Ten  Phone
-----
732  BENSON, E.T.           PHD   N   5156
218  HINCKLEY, G.B.        MBA   N   5509
854  KIMBALL, S.W.         PHD   Y   5594
-----
                LIST OF TEACHERS                Page 2
ID#           Name           Degr  Ten  Phone
-----
626  YOUNG, B.              MBA   Y   5664
574  SMITH, J.              MS    Y   5320

```

This is what we would see if we had used RECFM=FB. Note the carriage control characters in column 1.

**APPENDIX B
CONVERTING FROM PC/370 TO MVS/ESA**

B.10

1	LIST OF TEACHERS				Page	1
OID#	Name	Degr	Ten	Phone		
---	-----	----	---	-----		
732	BENSON, E.T.	PHD	N	5156		
218	HINCKLEY, G.B.	MBA	N	5509		
854	KIMBALL, S.W.	PHD	Y	5594		
1	LIST OF TEACHERS				Page	2
OID#	Name	Degr	Ten	Phone		
---	-----	----	---	-----		
626	YOUNG, B.	MBA	Y	5664		
574	SMITH, J.	MS	Y	5320		